
MODELLING INTER-REGIONAL TRADE FLOWS: 

DATA AND METHODOLOGICAL ISSUES IN RHOMOLO 

 

Olga Ivanova1, d’Artis Kancs2 and Dirk Stelder3 

 

 

Abstract: The present paper studies the modelling of inter-regional trade flows in the newly 

developed general equilibrium model of European Commission, Rhomolo. In contrast to 

international trade, usually, trade of goods and services between regions is not recorded 

statistically. As a result, no statistical data exists for inter-regional trade, which could be 

readily used for modelling purposes. We propose an augmented entropy approach, which 

combines the Transtools model’s inter-regional trade and inter-regional transport cost data 

with data from regional and national statistics to obtain a full set of balanced bilateral trade 

data for all inter-regional transactions. 

1 Introduction 

Regions and countries are linked with each other through trade of goods and services, factor, 

and income flows. Regions import and export goods and services from other regions in the 

same country and from other countries. The pattern of intensity of trade differs, however, 

between regions and their trading partners. 

In contrast to international trade, usually, trade of goods and services between regions is not 

recorded statistically. As a result, no statistical data exists for inter-regional trade, which 

could be readily used for modelling purposes. Such data, however, is necessary for empirical 

implementation of inter-regional general equilibrium models, such as Rhomolo. 

The objective of this paper is to discuss the data and methodological issues of modelling inter-

regional trade flows and, based on the gained insights, to develop a strategy for implementing 

the inter-regional trade in Rhomolo. 

 

2 Estimating inter-regional trade 

The intensity and pattern of trade between regions depends on many factors, such as, 

geographic and cultural proximity of the trading partners, consumer preferences, trade costs, 

and the structure and size of trading economies. The relative importance of each factor driving 

or hampering the inter-regional trade can be estimated econometrically, but as mentioned 

above, only for trade between countries. 

Conceptually, trade flows can be estimated in many different methodological and empirical 

specifications, which differ by the data requirements. Due to the easiness of implementation 

and its empirical success, offten, trade flows are modelled in a gravity framework. 

Different specifications and micro-foundations of the gravity model have been proposed in 

the literature (Feenstra 2002). For example, Anderson and van Wincoop (2003) show that in a 

one-sector economy, when consumers have CES preferences, trade between regions r and s 

can be expressed as: 

(1) XDDEr,s= (XDr*XDs/XD)*(trmr,s)1-σ/(MTRr*MTRs)1-σ 

where XDr and XDs are the levels of GDP, XD is world GDP, trmr,s are transport costs 

between r and s, and MTRr and MTRs are multilateral trade resistance terms of exporting and 

importing regions. They summarise the average trade resistance between a region and its 

trading partners in sense of ideal aggregation (Anderson and van Wincoop 2003). According 

to equation (1), the bilateral trade flows between r and s depend on bilateral costs, trmr,s, 

divided by the product of their multilateral resistance indices for s’s imports and for r’s 

exports. Hence, according to equation (1), the bilateral trade between r and s depends also on 

the relative trade barriers. 

In the gravity model of trade (1) all variables but the bilateral trade cost and the elasticity of 

substitution parameters are available in the data. These two parameters – trade costs and the 

elasticity of substitution – are not observable; usually they are estimated or calibrated. 

The use of equation (1) for the estimation of these two parameters is subject to several issues. 

First, the identification of trade costs, trmr,s, and the elasticity of substitution, σ, is not 

straightforward, as trade flows are decreasing in both the bilateral trade costs and the elasticity 

of substitution. As a result, the two missing parameters are usually calibrated. The 

identification issue in the estimation/calibration exercise is not important in the Rhomolo 

model, because a very detailed generalised transport cost data by region and sector are 

available. 

Second, the estimation of equation (1) requires data on inter-regional trade flows, XDDEr,s 

(bilateral flows for the base year or times series). However, as noted above, no statistical data 

exists for inter-regional trade flows, XDDEr,s, because firm sales are not recorded by region. 

Moreover, given that the upstream industry consists of many agents (wholesalers, transport, 

storage, retailers, etc), even producers cannot always track the spatial flow of their output. 

Again, this issue is not important for Rhomolo, because inter-regional trade flows are not used 

for model’s parameterisation. 

 

4 Modelling inter-regional trade in CGE 

4.1 The maximum entropy approach 

In general equilibrium models of trade, the trade of goods and services is determined 

endogenously as an optimisation outcome of the optimal strategy of firms, households and 

other economic 'actors'. Solving the model for the equilibrium prices and quantities, yields 

also inter-regional trade flows which reflect the underlying theoretical trade relationships and 

behavioural parameters. 

In a general equilibrium model the predicted trade flows need to satisfy also the 

macroeconomic data constraints for the base year. The macroeconomic constraints can be 

imposed e.g. via the maximum entropy method. The maximum entropy method allows to 

infer inter-regional trade flows such that, on the one hand, they satisfy the imposed data 

constraints and, on the other hand, the distance to the initial (unconstrained) estimates of trade 

flows is minimised. 

In the context of our study the maximum entropy approach has several advantages. First, the 

maximum entropy approach is flexible in sense that it allows to take into account different 

data sources for inter-regional trade flows as well as to put economic restrictions on the data, 

e.g. by the production and consumption accounts in regions. Second, the maximum entropy 

approach employs a nonlinear programming technique and, hence, can handle non-linear 

problems. Third, it can be implemented in GAMS, which is the programming language used 

in Rhomolo. 

4.2 Objective function 

The main component of the maximum entropy problem is the objective function. The 

objective function represents the distance between the estimated inter-regional trade flows, 

which satisfy the set of restrictions, and the initial (unrestricted) estimates of inter-regional 

trade flows. 

In the GAMS code (see Appendix) the objective function of the entropy problem is 

represented by EQentropy: 

entropy =E= sum((sec,reg,regg)$XDDE_o(sec,reg,regg),XDDE_V(sec,reg,regg) 

            *log(XDDE_V(sec,reg,regg)/XDDE_o(sec,reg,regg))) 

            + sum((sec,reg,regg)$transpm(sec,reg,regg),transpmV(sec,reg,regg) 

            *log(transpmV(sec,reg,regg)/transpm(sec,reg,regg))) 

            + sum((sec,cnt)$shRoEUtrade(cnt,sec),shRoEUtradeV(cnt,sec) 

            *log(shRoEUtradeV(cnt,sec)/shRoEUtrade(cnt,sec))) 

            + sum((sec,cnt)$shRoEUtradeE(cnt,sec),shRoEUtradeEV(cnt,sec) 

            *log(shRoEUtradeEV(cnt,sec)/shRoEUtradeE(cnt,sec))) 

 

4.3 Data constraints 

The second component of the maximum entropy problem is satisfying the constraints. One set 

of constrains represents the equality between demands for goods and services in each region 

and the total inflow of trade. Another set of constrains represents the equality between 

supplies of goods and services in each region and the total outflow of trade. Regional specific 

demands and supplies in each region are derived based on the regional production and 

consumption statistics. Another important constraint is that the aggregate inter-regional trade 

flows should be consistent with the inter-national trade statistics at the national level. Finally, 

regional trade cost margins should sum up to the transport services in national accounts. 

Hence, data on regional production; regional consumption; bilateral trade of goods and 

services between countries; and inter-regional trade costs (transport cost margins) provide 

constraints, which have to be satisfied when estimating inter-regional trade flows. These four 

constraints envelope the potential set of trade flows estimations. The more constraints can be 

derived from the theory and the available data, the smaller is the potential estimation error of 

inter-regional trade flows. 

4.3.1 Regional production 

The first constraint which needs to be satisfied is regional production. Region r produces 

goods of value XDr. The regional production constraint says that the total sales from region r, 

∑r,sXDDEr,s, to all regions (intra-regional plus inter-regional) must be equal to regional 

production, XDr. 

(2) XDr=∑sXDDEr,s 

For the purpose of Rhomolo, production data, XDr, by region and sector are readily available 

in statistics. 

A general GAMS notation of equation (2) is as follows: 

XDZ(sec,reg) =E= sum(regg,XDDE_V(sec,reg,regg)) 

In the GAMS code (see Appendix) equation (2) is represented by EQ11 and EQ12: 

XDZ(sec,reg)- EEU25Z(sec,reg) - EROWZ(sec,reg) =E= 

               sum(regg,XDDE_V(sec,reg,regg)) 

XDZ(sec,reg)  =E= sum(regg$XDDE_o(sec,reg,regg), XDDE_V(sec,reg,regg)) 

               + EEU25Z(sec,reg)*sum(mapCR(cnt,reg),shRoEUtradeEV(cnt,sec)) 

               + EROWZ(sec,reg) 

4.3.2 International trade flows 

The second constraint which needs to be satisfied is international trade flows. Part of goods 

and services produced in region r are sold in r, the rest is exported to other regions (within the 

same country and to other countries). The international trade constraint says that the total 

amount of goods and services shipped from all regions in a country to each foreign country 

must sum up to the international trade flows between the two countries. 

(3) XDDEc,cc=∑rXDDEr,s  r€c, s€cc 

For the purpose of Rhomolo, bilateral trade flows at country level, XDDEc,cc, are readily 

available in international trade statistics. 

In the GAMS code (see Appendix) equation (3) is represented by EQ3. 

sum((sec,reg)$mapCR(cnt,reg),XDDEZ(sec,reg)*scal) 

              =E=  sum(sec, sum((reg,regg)$mapCR(cnt,reg), 

              XDDE_V(sec,reg,regg))) 

4.3.3 Regional consumption 

Part of goods consumed, Xs, in region s stems from local production, the rest is imported from 

other regions. Both local sales (within the region) and inter-regional sales (to other regions) of 

all goods and services are subject to trade costs, trmr,s. Trade costs, trmr,s>0, increase the price 

of consumption goods, when they reach consumers. Hence, the transportation of goods 

XDDEr,s from region r increases goods’ value in destination region s to XDDEr,s*( 

PDCs+PTMr,s*trmr,s). 

The regional consumption constraint says that the value of goods consumed, Ps*Xs, in region 

s is equal to the total value of goods, ∑sXDDEr,s, sold in region s valued at producer prices 

and transportation costs, PDCs+PTMr,s*trmr,s. 

(4) Ps*Xs=∑sXDDEr,s*(PDCs+PTMr,s*trmr,s) 

For the purpose of Rhomolo, consumption data (value of goods consumed), Ps*Xs, by region 

and sector are readily available in statistics. 

The inter-regional trade cost data is available from the Transtools model, which provides 

bilateral trade costs by NUTS2 region and sector for the base year. In order to assure the 

quality of the Transtools data, the transportation margins from the Transtools model have 

undergone extensive checks of their quality, reliability and consistency. The new 

transportation cost data from the Transtools model satisfy all four criteria. Therefore, it is 

suitable for being used in the Rhomolo model. 

A general GAMS notation of equation (4) is as follows: 

XZ(sec,reg) =E= sum(regg,XDDE_V(sec,regg,reg)*(1+transpmV(sec,regg,reg))) 

In the GAMS code (see Appendix) equation (4) is represented by EQ21 and EQ22: 

XZ(sec,reg) - MEU25Z(sec,reg) - MROWZ(sec,reg) =E= 

               sum(regg, XDDE_V(sec,regg,reg)*(1+transpmV(sec,regg,reg))) 

XZ(sec,reg) =E= sum(regg$XDDE_o(sec,regg,reg), 

               XDDE_V(sec,regg,reg)*(1+transpmV(sec,regg,reg))) 

               + MEU25Z(sec,reg)*sum(mapCR(cnt,reg),shRoEUtradeV(cnt,sec)) 

               + MROWZ(sec,reg) 

4.3.4 Inter-regional trade costs 

In order to transport goods and services between and within regions, transport services need to 

be purchased. The trade cost constraint says that the sum of transport services purchased from 

all regions r in country c, ∑rXDDEr,s*trmr,s, is equal to the amount of transport services, 

TMXc, produced in country c. 

(5) TMXr=∑sXDDEr,s*trmr,s r€c 

For the purpose of Rhomolo, the total amount of transport services, TMXc, produced in 

country c is available from the national accounts data which enters the national Social 

Accounting Matrices. 

In the GAMS code (see Appendix) equation (5) is represented by EQ4: 

sum((sec,reg)$mapCR(cnt,reg),TMXZ(sec,reg)*scal) 

              =E= sum(sec,sum((reg,regg)$mapCR(cnt,reg), 

              XDDE_V(sec,reg,regg)*transpmV(sec,reg,regg))) 

 

4.4 The estimation procedure 

In a first step the gravity model of trade (1) is estimated. This yields unrestricted estimates of 

inter-regional trade flows. Given that no restrictions are imposed on the estimates at this 

stage, the unrestricted trade flows are not consistent with regional and national accounts. 

Alternatively, the unrestricted estimates of inter-regional trade flows can be taken directly 

from data, if available. In such a case, no gravity models need to be estimated. Given that the 

Transtools model provides data also on inter-regional trade flows, both approaches will be 

explored in Rhomolo. This will increase the robustness of model’s results. 

In a second step, in order to impose the consistency between trade flows and the rest of the 

model variables, the constraints (2)-(5) are imposed by applying the maximum entropy 

approach. This means that starting from the unrestricted trade flow estimates, the maximum 

entropy approach searches for the smallest deviations such that satisfying the constraints (2)-

(5): the regional production constraint; the international trade constraint; the regional 

consumption constraint; and the trade cost constraint. The obtained trade flow estimates are 

consistent with both the national and regional accounts data. 

 

5 Data sources 

5.1 Trade cost data 

The inter-regional trade cost data is available from the Transtools model, which provides 

bilateral trade costs by NUTS2 region and sector for the base year. In order to assure the 

quality of the Transtools data, the transportation margins from the Transtools model have 

undergone extensive checks of their quality, reliability and consistency. In particular, all 

bilateral trade cost margins have been analysed according to the following four criteria: 

• The share of transportation margin in the goods value. We expect that this margin 

would not exceed 0.2-0.3 for most types of goods and regions (Anderson and von 

Woncoop 2003). 

• The ratio of intra-regional trade costs to inter-regional trade costs. Normally, this ratio 

should be lower than 1.0 for most sectors and regions, because usually it is less costly 

to sell a good in the same region it is produced than to export it to other regions. 

• Symmetry of trade costs between trade flows of opposite direction for the same pair of 

regions. We expect the coefficient of correlation between the bilateral trade cost 

matrix and its inverse is larger than 0.5. 

• Trade cost outliers. We expect that the share of outliers within each sector and 

geographically proximate regions exceeding one standard deviation would not be 

larger than 0.1. 

The new transportation cost data from the Transtools model satisfy all four criteria. Therefore, 

it is suitable for being used in the Rhomolo model. 

Appendix: GAMS code of the entropy problem 

$TITLE:RHOMOLOENTROPY 

$eolcom # 

Option LIMROW = 0, LIMCOL = 0; 

option decimals = 8 ; 

 

Sets 

     b         numbers of columns and rows in the SAM /1*65/ 

     cact(b)   sectors + commodities - numeric        /1*46/ 

     cm(cact)  sectors - numeric - commodities        /1*23/ 

     act(cact) sectors - numeric - activities         /24*46/ 

     count     for reading in data                    /1*44/ 

     sec       sectors                                /sec1*sec23/ 

     pubs(sec) public sectors                         /sec10,sec20*sec23/ 

     transp(sec)    transport services                /sec15/ 

     goods(sec)     goods                             /sec1*sec9/ 

     cnt       countries                              /DE,PL,CZ,HU,SK/ 

     reg       regions                                / 

$include Data/NUTS2/PL.txt 

$include Data/NUTS2/DE.txt 

$include Data/NUTS2/HU.txt 

$include Data/NUTS2/SK.txt 

$include Data/NUTS2/CZ.txt 

     / 

     regPL(reg)    Polish regions 

     / 

$include Data/NUTS2/PL.txt 

     / 

     regDE(reg)    German regions 

     / 

$include Data/NUTS2/DE.txt 

     / 

     regHU(reg)    Hungarian regions 

     / 

$include Data/NUTS2/HU.txt 

     / 

     regSK(reg)    Slovak regions 

     / 

$include Data/NUTS2/SK.txt 

     / 

     regCZ(reg)    Czech regions 

     / 

$include Data/NUTS2/CZ.txt 

     / 

 

     mapCR(cnt,reg) mapping between NUTS regions and countries 

     mapRR(reg,reg) mapping between regions of the same country 

; 

 

mapCR("DE",reg)$regDE(reg) = yes ; 

mapCR("PL",reg)$regPL(reg) = yes ; 

mapCR("HU",reg)$regHU(reg) = yes ; 

mapCR("SK",reg)$regSK(reg) = yes ; 

mapCR("CZ",reg)$regCZ(reg) = yes ; 

 

Alias 

         (b,bb) 

         (cact,cactt) 

         (cm,cmm) 

         (act,actt) 

         (reg,regg) 

         (reg,reggg) 

         (reg,regggg) 

         (sec,secc) 

         (sec,seccc) 

         (cnt,cntt) 

         (cnt,cnttt) 

; 

 

Parameters 

   XZ(sec,reg)           initial domestic sales 

   XDZ(sec,reg)          initial domestic gross production (output) level 

   TMXZ(sec,reg)         initial consumption for prod of transport margins 

   EEU25Z(sec,reg)       initial exports to EU25 

   EROWZ(sec,reg)        initial exports to RoW 

   MEU25Z(sec,reg)       initial imports from EU25 

   MROWZ(sec,reg)        initial imports from RoW 

   shRoEUtrade(cnt,sec)  share of RoEU trade in total EU imports 

   shRoEUtradeE(cnt,sec) share of RoEU trade in total EU exports 

   XDDEZ(sec,reg)        international trade between countries 

 

   XDDE_o(sec,reg,regg)  initial split according to origin indic (TRANSTOOLS) 

   transpm(sec,reg,regg) transport margins (TRANSTOOLS) 

; 

 

$libinclude xlimport XZ Data/Entropydata.xls XZ! ; 

$libinclude xlimport XDZ Data/Entropydata.xls XDZ! ; 

$libinclude xlimport TMXZ Data/Entropydata.xls TMXZ! ; 

$libinclude xlimport EEU25Z Data/Entropydata.xls EEU25Z! ; 

$libinclude xlimport EROWZ Data/Entropydata.xls EROWZ! ; 

$libinclude xlimport MEU25Z Data/Entropydata.xls MEU25Z! ; 

$libinclude xlimport MROWZ Data/Entropydata.xls MROWZ! ; 

$libinclude xlimport shRoEUtrade Data/Entropydata.xls shRoEUtrade! ; 

$libinclude xlimport shRoEUtradeE Data/Entropydata.xls shRoEUtradeE! ; 

$libinclude xlimport XDDEZ Data/Entropydata.xls XDDEZ! ; 

 

$libinclude xlimport XDDE_o Data/TranstoolsData.xls Tonnes!C1..O35926 ; 

$libinclude xlimport transpm Data/TranstoolsData.xls GCostUnit!C1..O35926 ; 

 

Scalar 

scal scaling parameter monetary values 

; 

scal = 1000 ; 

Positive variables 

XDDE_V(sec,reg,regg) 

transpmV(sec,regg,reg) 

shRoEUtradeV(cnt,sec) 

shRoEUtradeEV(cnt,sec) 

; 

Variables 

entropy 

; 

 

Equations 

EQ11(sec,reg) 

EQ12(sec,reg) 

EQ21(sec,reg) 

EQ22(sec,reg) 

EQ3(cnt) 

EQ4(cnt) 

EQentropy 

; 

 

EQ11(sec,reg)$((XDZ(sec,reg)- EEU25Z(sec,reg) - EROWZ(sec,reg)) and (not goods(sec)) ).. 

               (XDZ(sec,reg)- EEU25Z(sec,reg) - EROWZ(sec,reg)) =E= 

               sum(regg, XDDE_V(sec,reg,regg)) ; 

 

EQ12(sec,reg)$(XDZ(sec,reg) and goods(sec)).. 

               XDZ(sec,reg)  =E= sum(regg$XDDE_o(sec,reg,regg), XDDE_V(sec,reg,regg)) 

               + EEU25Z(sec,reg)*sum(mapCR(cnt,reg),shRoEUtradeEV(cnt,sec)) 

               + EROWZ(sec,reg) ; 

 

EQ21(sec,reg)$((XZ(sec,reg) - MEU25Z(sec,reg) - MROWZ(sec,reg)) and (not goods(sec)) ).. 

               (XZ(sec,reg) - MEU25Z(sec,reg) - MROWZ(sec,reg)) =E= 

               sum(regg, XDDE_V(sec,regg,reg)*(1+transpmV(sec,regg,reg))) ; 

 

EQ22(sec,reg)$(XZ(sec,reg) and goods(sec)).. 

               XZ(sec,reg) =E= sum(regg$XDDE_o(sec,regg,reg), 

               XDDE_V(sec,regg,reg)*(1+transpmV(sec,regg,reg))) 

               + MEU25Z(sec,reg)*sum(mapCR(cnt,reg),shRoEUtradeV(cnt,sec)) 

               + MROWZ(sec,reg) ; 

 

EQ3(cnt)..   sum((sec,reg)$mapCR(cnt,reg),XDDEZ(sec,reg)*scal) 

              =E=  sum(sec, sum((reg,regg)$mapCR(cnt,reg), 

              XDDE_V(sec,reg,regg))) ; 

 

EQ4(cnt)..   sum((sec,reg)$mapCR(cnt,reg),TMXZ(sec,reg)*scal) 

              =E=  sum(sec, sum((reg,regg)$mapCR(cnt,reg), 

              XDDE_V(sec,reg,regg)*transpmV(sec,reg,regg))) ; 

 

EQentropy.. entropy =E= sum((sec,reg,regg)$XDDE_o(sec,reg,regg),XDDE_V(sec,reg,regg) 

            *log(XDDE_V(sec,reg,regg)/XDDE_o(sec,reg,regg))) 

            + sum((sec,reg,regg)$transpm(sec,reg,regg),transpmV(sec,reg,regg) 

            *log(transpmV(sec,reg,regg)/transpm(sec,reg,regg))) 

            + sum((sec,cnt)$shRoEUtrade(cnt,sec),shRoEUtradeV(cnt,sec) 

            *log(shRoEUtradeV(cnt,sec)/shRoEUtrade(cnt,sec))) 

            + sum((sec,cnt)$shRoEUtradeE(cnt,sec),shRoEUtradeEV(cnt,sec) 

            *log(shRoEUtradeEV(cnt,sec)/shRoEUtradeE(cnt,sec))) ; 

 

Model entropy_min /EQ11,EQ12,EQ21,EQ22,EQ3,EQ4,EQentropy/ ; 

 

XDDE_V.L(sec,reg,regg)   =  XDDE_o(sec,reg,regg) ; 

transpmV.L(sec,reg,regg) =  transpm(sec,reg,regg); 

shRoEUtradeV.L(cnt,sec)  =  shRoEUtrade(cnt,sec) ; 

shRoEUtradeEV.L(cnt,sec) =  shRoEUtradeE(cnt,sec) ; 

 

XDDE_V.FX(sec,reg,regg)$(XDDE_o(sec,reg,regg) eq 0)     =  XDDE_o(sec,reg,regg) ; 

transpmV.FX(sec,reg,regg)$(transpm(sec,reg,regg) eq 0)  =  transpm(sec,reg,regg); 

shRoEUtradeV.FX(cnt,sec)$(shRoEUtrade(cnt,sec) eq 0)    =  shRoEUtrade(cnt,sec) ; 

shRoEUtradeEV.FX(cnt,sec)$(shRoEUtradeE(cnt,sec) eq 0)  =  shRoEUtradeE(cnt,sec); 

 

Option iterlim = 1000000 ; 

Option reslim = 1000000 ; 

option nlp = conopt3 ; 

 

Solve entropy_min using nlp minimizing entropy ; 

 

Display XDDE_V.L, transpmV.L,shRoEUtradeEV.L,shRoEUtradeV.L ; 

 

loop ((sec,reg), 

loop(regg, 

   if ((XDDE_V.L(sec,reg,regg) lt 0), 

        abort "check XDDE(sec,reg,regg)" 

   ); 

); 

); 


